

# Index notation in Lean 4

Joseph Tooby-Smith

*Reykjavik University, Reykjavik, Iceland*

November 1, 2024

## Abstract

The physics community relies on index notation to effectively manipulate tensors of specific types. This paper introduces the first formally verified implementation of index notation in the interactive theorem prover Lean 4. By integrating index notation into Lean, we bridge the gap between traditional physics notation and formal verification tools, making it more accessible for physicists to write and prove results within Lean. In the background, our implementation leverages a novel application of category theory.

The file you are currently reading is a draft. There will be typos, mistakes, sentences that don't make sense, notation that is not defined etc.

## 1. INTRODUCTION

In previous work, the author initiated the digitalization (or formalization) of high energy physics results using the interactive theorem prover Lean 4 in a project called HepLean. Lean is a programming language with syntax resembling traditional pen-and-paper mathematics. Users can write definitions, theorems, and proofs in Lean, which are then automatically checked using dependent-type theory for correctness. The HepLean project is driven by four primary motivations: to facilitate easier look-up of results through linear information storage; to support the creation and proof of new results using automated tactics and AI tools; to facilitate correctness checking of result in high energy physics; and to introduce new pedagogical methods for high-energy physics and computer science.

HepLean is part of a broader movement of projects to formalize parts, or all, of mathematics and science. The largest of these projects is Mathlib, which aims to formalize mathematics. Indeed, HepLean is built downstream of Mathlib, meaning it has Mathlib as a dependency and uses many of the definitions and theorems thereof. Other projects include the ongoing effort led by Kevin Buzzard to formalize the proof of Fermat's Last Theorem into Lean. In the realm of the sciences, the paper [?] looks at absorption theory, thermodynamics, and kinematics in Lean. Whilst the package SciLean [?], is a push in the direction of scientific computing within Lean.

Physicists heavily rely on specialized notation to express mathematical concepts succinctly. Among these, index notation is particularly prevalent, as it provides a compact and readable way to represent specific types of tensors and operations between them, with tensors forming a backbone of modern physics.

Because of the importance of index notation, it is crucial to have a way to write index notation in Lean. This will make results from high energy physics easier to write and prove in Lean. Additionally, it will make the syntax more familiar to high energy physicists. Such an implementation is the subject of this paper. It is a challenge because, not only do we need an implementation that is nice and flexible to use,



Figure 1: Overview of the implementation of index notation in Lean. The solid lines represent formally verified parts of the implementation.

we also need it to be formally well-defined and rigorous. We hope that the implementation presented here will not only enhance usability of Lean but also promotes the adoption of formal methods in the physics community.

As a taster of what is to come, our implementation allows to write results like the following in Lean:

```
{pauliCo | v α β ⊗ pauliContr | v α' β' = 2 · εL | α α' ⊗ εR | β β'}T
```

Lean will correctly interpret this result as a tensor expression which the correct contraction of indices and permutation of indices between the sides of the expression.

Previous implementations of index notation have been made in programming languages like Haskell [?]. However, these implementations do not provide the formal verification capabilities inherent in Lean. The formal verification requirement of Lean introduces unique challenges in implementing index notation, necessitating (what we believe is) a novel solution.

This paper is split into two main sections. Section 2 discusses the implementation of index notation into Lean. Section [js: ref](#) gives two examples of theorems and proofs using index notation. A more minor section, section [js: ref](#) discusses the future work related to this project.

## 2. IMPLEMENTATION OF INDEX NOTATION INTO LEAN 4

Our implementation of index notation in Lean can be broken down into three main components, illustrated in Figure 1. The first component is the *syntax for tensor expressions*, which is what users interact with when writing results in Lean. This syntax closely mirrors the notation familiar to physicists, making it intuitive and accessible. It appears directly in the Lean files and can be thought of as an informal string that represents the tensor expressions. The code snippet above, involving pauli matrices, is an example of this.

The second component involves transforming this syntax into a *tensor tree*. The tensor tree is a formal mathematical representation of the tensor expression. By parsing the syntax into a structured tree, we establish a rigorous foundation that captures the tensor expression. This formal representation allows us to easily manipulate tensor expressions and prove results related to them in a way that Lean accepts as formal.

The third and final component is the conversion of the tensor tree into an actual *tensor*. This process utilizes properties of the symmetric-monoidal category of representations to translate the tensor tree into a tensor.

Although Lean processes information from left to right in Figure 1, starting with the syntax and proceeding to the tensor tree, and then finally (when we ask it to) to the tensors, it is more effective to discuss the implementation from right to left. Starting with tensors is advantageous because they are the primary objects of interest. The left and middle parts of the diagram can be thought of as intermediate stages that facilitate the manipulation and understanding of these tensors and their expressions.

## 2.1. DEFINING TENSORS

### 2.1.1 Building blocks of tensors and color

Tensors of a species, such as complex Lorentz tensors, are constructed from a set of building block representations of a group  $G$  over a field  $k$ .

For complex Lorentz tensors, the group  $G$  is  $SL(2, \mathbb{C})$ , the field  $k$  is the field of complex numbers and six building block representations. The six building block representations are

- The representation of left-handed Weyl fermions, denoted in Lean as `Fermion.leftHanded`, and corresponding to the representation of  $SL(2, \mathbb{C})$  taking  $v \mapsto Mv$  for  $M \in SL(2, \mathbb{C})$ .
- The representation of ‘alternative’ left-handed Weyl fermions, denoted in Lean as `Fermion.altLeftHanded`, and corresponding to the representation of  $SL(2, \mathbb{C})$  taking  $v \mapsto M^{-1T}v$  for  $M \in SL(2, \mathbb{C})$ .
- The representation of right-handed Weyl fermions, denoted in Lean as `Fermion.rightHanded`, and corresponding to the representation of  $SL(2, \mathbb{C})$  taking  $v \mapsto M^*v$  for  $M \in SL(2, \mathbb{C})$ .
- The representation of ‘alternative’ right-handed Weyl fermions, denoted in Lean as `Fermion.altRightHanded`, and corresponding to the representation of  $SL(2, \mathbb{C})$  taking  $v \mapsto M^{-1\dagger}v$  for  $M \in SL(2, \mathbb{C})$ .
- The representation of contravariant Lorentz tensors, denoted in Lean as `Lorentz.complexContr`, and corresponding to the representation of  $SL(2, \mathbb{C})$  induced by the homomorphism of  $SL(2, \mathbb{C})$  into the Lorentz group and the contravariant action of the Lorentz group on four-vectors.
- The representation of covariant Lorentz tensors, denoted in Lean as `Lorentz.complexCo`, and corresponding to the representation of  $SL(2, \mathbb{C})$  induced by the homomorphism of  $SL(2, \mathbb{C})$  into the Lorentz group and the covariant action of the Lorentz group on four-vectors.

An example of how these representations are defined in Lean is given below for the representation of left-handed Weyl fermions, `Fermion.leftHanded`:

```

/-- The vector space  $\mathbb{C}^2$  carrying the fundamental representation of  $SL(2, \mathbb{C})$ .
In index notation corresponds to a Weyl fermion with indices  $\psi^a$ . -/
def leftHanded : Rep  $\mathbb{C}$   $SL(2, \mathbb{C})$  := Rep.of {
  /- The function from  $SL(2, \mathbb{C})$  to endomorphisms of LeftHandedModule
  (which corresponds to the vector space  $\mathbb{C}^2$ ). -/
  toFun := fun M => {
    /- Start of the definition of the linear map. -/
    /- The function underlying the linear map. Defined as the dot product. -/
    toFun := fun ( $\psi$  : LeftHandedModule) =>
      LeftHandedModule.toFin2CEquiv.symm (M.1 *v  $\psi$ .toFin2C),
    /- Proof that the function is linear with respect to addition. -/
    map_add' := by
      intro  $\psi$   $\psi'$ 
      simp [mulVec_add]
    /- Proof that the function is linear with respect to scalar multiplication. -/
    map_smul' := by
      intro r  $\psi$ 
      simp [mulVec_smul]
    /- End of the definition of the linear map. -/
  }

```

```

/- Proof that (the outer) toFun gives the identity map on the identity of
   SL(2,C). -/
map_one' := by
  ext i
  simp
/- Proof that the action of the product of two elements is
   the product of the actions of the elements. -/
map_mul' := fun M N => by
  simp only [SpecialLinearGroup.coe_mul]
  ext1 x
  simp only [LinearMap.coe_mk, AddHom.coe_mk, LinearMap.mul_apply,
    LinearEquiv.apply_symm_apply,
    mulVec_mulVec]}

```

We have added some explanatory comments to this code, not seen in the actual Lean code, to give the reader an idea of what each part does.

Moving on, to each of the building block representations, we assign a unique label, which we refer to as a *color*. We then get a type of colors for a given species of tensors, which in this paper we will denote  $C$ .

For complex Lorentz tensors we associate the color upL to the representation `Fermion.leftHanded`, downL to `Fermion.altLeftHanded`, upR to `Fermion.rightHanded`, downR to `Fermion.altRightHanded`, up to `Lorentz.complexContr` and down to `Lorentz.complexCo`. The type of colors is defined in Lean as an inductive type as follows:

```

inductive Color
| upL : Color
| downL : Color
| upR : Color
| downR : Color
| up : Color
| down : Color

```

This can roughly be thought of as the type with six elements. We will see a much more complicated form of inductive type when we come to define tensor trees.

The assignment of colors to their corresponding representations is done via a functor  $F_D$  from the set  $C$  to the category of  $k$ -representations of  $G$ , which we denote  $\text{Rep } k G$ . That, is a functor

$$F_D : C \Rightarrow \text{Rep } k G. \quad (1)$$

Since the source category of this functor is just a set, we could have instead defined it as a function - however we will see later that we want to lift this functor to a symmetric monoidal functor with a more sophisticated source category. For this reason, it is useful to have a functor here.

However, been a functor from a set, means that defining it for specific cases in Lean is somewhat stright forard. For complex Lorentz tensors this functor is defined as follows:

```

FDiscrete := Discrete.functor fun c =>
  match c with
  | Color.upL => Fermion.leftHanded
  | Color.downL => Fermion.altLeftHanded
  | Color.upR => Fermion.rightHanded
  | Color.downR => Fermion.altRightHanded
  | Color.up => Lorentz.complexContr
  | Color.down => Lorentz.complexCo

```

This code snippet actually forms part of a larger structure called a tensor species, which we will get on to shortly.

## 2.1.2 General tensors

Given any type  $C$ , which we will think of as a type of colors, we can define the category  $T_{/C}^\times$  as follows. Objects are functions  $f : X \rightarrow C$  for some type  $X$ . A morphism from  $f : X \rightarrow C$  to  $g : Y \rightarrow C$  is a bijection  $\phi : X \rightarrow Y$  such that  $f = g \circ \phi$ . This category is equivalent to the core of the category of types sliced over  $C$ , which explains the origin of the notation. In Lean we denote this category as `OverColor C`, although we will use the shorter notation  $T_{/C}^\times$  here when we can.

The category  $T_{/C}^\times$  is not any old category, it carries a symmetric monoidal structure, which we will denote  $\otimes$ . The structure such that  $f \otimes g$  for objects  $f$  and  $g$  is the map  $X \oplus Y \rightarrow C$  where  $\oplus$  denotes the disjoint union of types.

The functor  $F_D$  defined above, can be lifted to a symmetric monoidal functor  $F$  from  $T_{/C}^\times$  to  $\text{Rep } k G$ . This functor takes  $f : X \rightarrow C$  to the tensor product  $\bigotimes_{x \in X} F_D(f(x))$  and morphisms to the linear maps of representations corresponding to reindexings of tensor products. This construction is general, and functorial. In other words, there is a functor

$$\text{lift} : \text{Fun}(C, \text{Rep } k G) \Rightarrow \text{SymmMonFun}(T_{/C}^\times, \text{Rep } k G) \quad (2)$$

In Lean we define this functor, along with the corresponding lift of  $F_D$ ,  $F$ .

We can think of an object  $f : X \rightarrow C$  of  $T_{/C}^\times$  as a type of indices  $X$ , and a specification of what representation each index is associated to. The representation  $F(f)$  is then the tensor product of each of these representations, so that vectors  $v \in F(f)$  can be thought of as tensors with indices indexed by  $X$  of color  $C$  (e.g. ‘up’ or ‘down’).

With this in mind, we define a general tensor of a given species as a vector in the representation  $F(f)$  for some  $f$  in  $T_{/C}^\times$ .

In physics, we typically focus on objects  $f : X \rightarrow C$  in  $T_{/C}^\times$  where  $X$  is a finite type of the form  $\text{Fin } n$  for some  $n$ . Here,  $\text{Fin } n$  represents the type of natural numbers less than  $n$ , i.e.,  $\{0, 1, \dots, n-1\}$ . When appropriate in what follows, we will restrict to these objects.

In Lean for a map of types  $f : X \rightarrow C$ , what we have been writing as  $F(f)$  is written as `S.F.obj (OverColor.mk f)`. The object `S` is the tensor species, which we will define shortly. The `OverColor.mk` tells lean that the function `f` should be considered as an object of `OverColor C`, not just as an object of type  $X \rightarrow C$ . [js: fix arrows here](#)

### 2.1.3 Tensor operations

There are eight fundamental operations one might wish to perform on tensors: addition, scalar multiplication, negation, action by the group  $G$ , permutation of indices, tensor product, contraction of indices, and evaluation of an index. In this section, we explore how each of these operations arises within the framework we have established so far. The operations we will discuss will reappear in Section ??, where we dicuss how to go from a tensor tree to a tensor.

The first three operations—addition, scalar multiplication, and negation—directly result from  $F(f)$  being a vector space over  $k$ . Consequently, in Lean, these structures are inherently available. Similarly, since  $F(f)$  resides in  $\text{Rep } k \ G$ , it naturally comes equipped with a group action, which is again inherently available to us in Lean.

Permuting indices of a tensors, or altering their indexing sets, can be achived by applying  $F$  to morphisms in  $T_{/C}^\times$ . Essentially, the morphisms in  $T_{/C}^\times$  can be thought of as valid permutations or reindexings of indices. They are valid in the sense that they will never interchange an ‘up’ index with a ‘down’ index, for instance.

Next, we consider the tensor product of two tensors. The tensor product of vectors is already defined in Mathlib. Using this, if we have  $t \in F(f)$  and  $s \in F(g)$ , we can form a vector in  $F(f) \otimes F(g)$ . In Lean, this is denoted as `js: sorry`. Our goal is to obtain some elemnet in some  $F\_$ , which we can achieve using the tensorate of  $F$ . Recall that  $F$  is a symmetric monoidal functor, and its tensorate is an isomorphism between  $F(f) \otimes F(g)$  and  $F(f \otimes g)$ .

We now turn to contraction, where we will also discuss the metric and unit. To define contraction we need to know what indices can be contracted with what others. For this we introduce an involution  $\tau : C \rightarrow C$ . For complex Lorentz tensors this is defined as follows

```
τ := fun c =>
  match c with
  | Color.upL => Color.downL
  | Color.downL => Color.upL
  | Color.upR => Color.downR
  | Color.downR => Color.upR
  | Color.up => Color.down
  | Color.down => Color.up
```

For a color  $c \in C$  we say that  $\tau(c)$  is the dual of  $C$ . What this will mean is that we can contract indices of color  $c$  with indices of color  $\tau(c)$ . For a color  $c$  the contraction corresponding to that color is an equivariant map from  $F(c) \otimes F(\tau c)$  to the trivial representation, which in this paper we will denote  $\mathbb{1}$ . That is to say, a morphism  $\mathcal{C}(c)$  in  $\text{Rep } k \ G$  from  $F(c) \otimes F(\tau c)$  to the unit of the monoidal structure, which is the trivial representation. Collectively, and somewhat trivially,  $\mathcal{C}$  forms a natural transformation from the functor  $F\_ \otimes F(\tau\_)$  to the constant functor landing on  $\mathbb{1}$ , which we will also denote  $\mathbb{1}$ .

We expect that the contraction  $\mathcal{C}(c)$  and the contraction  $\mathcal{C}(\tau(c))$  are essentially the same. After all, for four-vectors one is the contraction of an up index with a down index whilst the other is the contraction of a down index with an up index. To make this formal we put a condition on  $\mathcal{C}$  which can

succinctly be written as a diagram:

$$\begin{array}{ccc}
 F_D(c) \otimes F_D(\tau c) & \xrightarrow{\beta} & F_D(\tau c) \otimes F_D(c) \\
 \mathcal{C}(c) \downarrow & & \downarrow \mathbb{1} \otimes F_D \tau_c \\
 \mathbb{1} & \xleftarrow{\mathcal{C}(\tau(c))} & F_D(\tau c) \otimes F_D(\tau \tau c)
 \end{array} \quad (3)$$

where  $\beta$  is the braiding of the symmetric monoidal category, and  $\tau_c$  is the isomorphism in  $C$  between  $\tau(\tau(c))$  and  $c$ , which exists since  $\tau$  is an involution. In Lean this condition is written in a easier-to-prove but equivalent way, by acting the relevant maps on pure vectors within  $F_D(c) \otimes F_D(\tau c)$ . We will see this in the next subsection.

The contraction we have defined so far does not specify how to contract indices of a general tensor in  $F(f)$ . Suppose we wish to contract the indices  $i$  and  $j$  in  $X$ , and that their colors allow such a contraction, i.e.,  $f(j) = \tau(f(i))$ . To achieve this, we establish an equivalence in  $T_{/C}^\times$  between  $f$  and  $![fi] \otimes ![fj]! \otimes f'$ , where  $![fi] : \text{Fin } 1 \rightarrow C$  lands on  $fi$  and similarly for  $fj$ , and  $f'$  can be thought of as  $f$  with  $i$  and  $j$  removed.

Using the tensorate of  $F$  and the equivalence just discussed, we obtain an equivalence between  $F(f)$  and  $(F(f(i)) \otimes F(f(j))) \otimes F(f')$ . This in turn is equivalent to  $(F_D(f(i)) \otimes F_D(\tau(f(i)))) \otimes F(f')$ . The contraction morphism  $\mathcal{C}(f(i))$  can then be applied to the first part of this expression, reducing it to a vector in  $\mathbb{1} \otimes F(f')$ . Finally, we can use the left unitor of  $\text{Rep } k G$  to obtain  $F(f')$ .

$$F f \equiv (F_D(fi) \otimes F_D(\tau(fi))) \otimes F(f'i) \rightarrow \mathbb{1} \otimes F(f'i) \rightarrow F(f'i) \quad (4)$$

js: Maybe restrict to *Finn* here.

We have some conditions on contraction. Firstly, we expect there to be a unit. That is a, group-invariant element of  $F(\tau(c)) \otimes F(c)$  which when contracted with does nothing. A group invariant element of  $F_D(\tau(c)) \otimes F_D(c)$  is the same thing as a morphism from  $\delta(c) : \mathbb{1} \rightarrow F_D(\tau(c)) \otimes F_D(c)$ . Again we can lift  $\delta$  to a natural transformation from  $\mathbb{1}$  to  $F(\tau_-) \otimes F_-$ . The condition that it does nothing when contracted with is that the following diagram commutes:

$$\begin{array}{ccccc}
 F_D(c) & \xrightarrow{\rho^{-1}} & F_D(c) \otimes \mathbb{1} & \xrightarrow{\mathbb{1} \otimes \delta(c)} & F_D(c) \otimes (F_D(\tau c) \otimes F_D(c)) \\
 \lambda \uparrow & & & & \downarrow \alpha^{-1} \\
 \mathbb{1} \otimes F_D(c) & \xleftarrow{\mathcal{C}(c) \otimes \mathbb{1}} & & & (F_D(c) \otimes F_D(\tau c)) \otimes F_D(c)
 \end{array} \quad (5)$$

where  $\rho$  is the right-unitor,  $\lambda$  the left-unitor, and  $\alpha$  is the associator. Again in Lean, we write this condition in an easier-to-prove but equivalent way.

We want these units to be symmetric, that is satisfy the condition that

$$\begin{array}{ccc}
 \mathbb{1} & \xrightarrow{\delta(c)} & F_D(\tau c) \otimes F_D(c) \\
 \downarrow \delta(\tau(c)) & & \uparrow \mathbb{1} \otimes F_D \tau_c \\
 F_D(\tau \tau c) \otimes F_D(\tau c) & \xrightarrow{\beta} & F_D(\tau c) \otimes F_D(\tau \tau c)
 \end{array} \quad (6)$$

Lastly, before we turn to evaluation, we want to have metrics. That is a group-invariant element of  $F_D(c) \otimes F_D(c)$  which when contracted together give the unit. As above, a group-invariant element of  $F_D(c) \otimes F_D(c)$  is the same thing as a morphism from  $\eta(c) : \mathbb{1} \rightarrow F_D(c) \otimes F_D(c)$ . We can treat  $\eta$  to a



natural transformation from  $\mathbb{1}$  to  $F_{D-} \otimes F_{D-}$ . The condition that it gives the unit when contracted together is that the following diagram commutes:

$$\begin{array}{ccc}
 \mathbb{1} & \xrightarrow{\delta(c)} & F_D(\tau c) \otimes F_D(c) \\
 \downarrow \eta(c) \otimes \eta(\tau(c)) & & \uparrow \beta \\
 (F_D c \otimes F_D c) \otimes (F_D(\tau c) \otimes F_D(\tau c)) & & F_D(c) \otimes F_D(\tau c) \\
 \downarrow & & \uparrow \mathbb{1} \otimes (\mathcal{E}(c) \otimes \mathbb{1}) \\
 F_D c \otimes (F_D c \otimes (F_D(\tau c) \otimes F_D(\tau c))) & \longrightarrow & F_D c \otimes ((F_D c \otimes F_D(\tau c)) \otimes F_D(\tau c))
 \end{array} \tag{7}$$

These metrics ensure that the representations  $F_D(c)$  and  $F_D(\tau(c))$  are actually equivalent to one another, and establish an isomorphism in  $\text{Rep } k G$  between them.

Let us turn to our last operation, evaluation of an index. This is perhaps the least important of the operations, not least because it is not equivariant with respect to the group action, like all of the other operations are. However, it is used often in physics so we include it here. To define evaluation, we give each  $F_D(c)$  a basis  $\{e_j^c\}$ . For each  $e_j^c$  we get a morphism in the category of vector spaces from  $F_D(c) \rightarrow k$  corresponding to taking the coordinate of a vector with respect to  $e$ . This can be made into an evaluation map for  $F(f)$  in a similar way to how we did contraction. We take an  $i$  in  $X$  and define an equivalence  $f \equiv ![fi] \otimes f'$ , formed by removing  $i$  from  $X$ . Using the tensorate of  $F$  we get an equivalence between  $F(f)$  and  $F_D(f(i)) \otimes F(f')$ . Moving to the category of vector spaces, we can use evaluation for our chosen  $e_j^{f(i)}$  to get a map from  $F_D(f(i)) \otimes F(f')$  to  $\mathbb{1} \otimes F(f')$ . Finally, we can use the left unitor to get a map from  $F(f)$  to  $F(f')$ . Putting this together we get a morphism, in the category of vector spaces

$$Ff \equiv F_D(f(i)) \otimes F(f') \rightarrow \mathbb{1} \otimes F(f') \rightarrow F(f') \tag{8}$$

#### 2.1.4 Tensor Species

js: In rewrite - I got up to here.

The data of the field  $k$ , the group  $G$ , the functor  $F_D$  (from which  $F$  can be derived), the involution  $\tau$ , the natural transformations for contraction, metric, and unit, and basis needed for evaluation, form formally what we have loosely been calling a Tensor species.

That is, the difference between complex Lorentz tensors, Einstein tensors, and real Lorentz tensors is down to this data. In Lean it useful to work with general tensor species where possible, so important results have to be defined only ones. Thus we make the following definition

```

/-- The structure of a type of tensors e.g. Lorentz tensors, Einstein tensors,
    complex Lorentz tensors. -/
structure TensorSpecies where
  /-- The colors of indices e.g. up or down. -/
  C : Type
  /-- The symmetry group acting on these tensor e.g. the Lorentz group or SL(2,C).
      -/
  G : Type
  /-- An instance of 'G' as a group. -/
  G_group : Group G
  /-- The field over which we want to consider the tensors to live in, usually 'R'
      or 'C'. -/

```



```

k : Type
/-- An instance of 'k' as a commutative ring. -/
k_commRing : CommRing k
/-- A 'MonoidalFunctor' from 'OverColor C' giving the rep corresponding to a map
of colors
'X → C'. -/
FDiscrete : Discrete C ⇒ Rep k G
/-- A map from 'C' to 'C'. An involution. -/
τ : C → C
/-- The condition that 'τ' is an involution. -/
τ_involution : Function.Involutive τ
/-- The natural transformation describing contraction. -/
contr : OverColor.Discrete.pair τ FDiscrete τ →  $\mathbb{1}_$  (Discrete C ⇒ Rep k G)
/-- The natural transformation describing the metric. -/
metric :  $\mathbb{1}_$  (Discrete C ⇒ Rep k G) → OverColor.Discrete.pair FDiscrete
/-- The natural transformation describing the unit. -/
unit :  $\mathbb{1}_$  (Discrete C ⇒ Rep k G) → OverColor.Discrete.τPair FDiscrete τ
/-- A specification of the dimension of each color in C. This will be used for
explicit
evaluation of tensors. -/
repDim : C → ℕ
/-- repDim is not zero for any color. This allows casting of 'ℕ' to 'Fin
(S.repDim c)'. -/
repDim_neZero (c : C) : NeZero (repDim c)
/-- A basis for each Module, determined by the evaluation map. -/
basis : (c : C) → Basis (Fin (repDim c)) k (FDiscrete.obj (Discrete.mk c)).V
/-- Contraction is symmetric with respect to duals. -/
contr_tmul_symm (c : C) (x : FDiscrete.obj (Discrete.mk c))
(y : FDiscrete.obj (Discrete.mk (τ c))) :
(contr.app (Discrete.mk c)).hom (x  $\otimes_{[k]}$  y) = (contr.app (Discrete.mk (τ
c))).hom
(y  $\otimes_{[k]}$  (FDiscrete.map (Discrete.eqToHom (τ_involution c).symm)).hom x)
/-- Contraction with unit leaves invariant. -/
contr_unit (c : C) (x : FDiscrete.obj (Discrete.mk (c))) :
(λ_ (FDiscrete.obj (Discrete.mk (c)))) .hom .hom
(((contr.app (Discrete.mk c)) ▷ (FDiscrete.obj (Discrete.mk (c)))).hom
((α_ _ (FDiscrete.obj (Discrete.mk (c)))).inv .hom
(x  $\otimes_{[k]}$  (unit.app (Discrete.mk c)).hom (1 : k)))) = x
/-- The unit is symmetric. -/
unit_symm (c : C) :
((unit.app (Discrete.mk c)).hom (1 : k)) =
((FDiscrete.obj (Discrete.mk (τ (c)))).hom
(FDiscrete.map (Discrete.eqToHom (τ_involution c)))).hom
((β_ (FDiscrete.obj (Discrete.mk (τ (τ c)))) (FDiscrete.obj (Discrete.mk (τ
(c)))).hom .hom
((unit.app (Discrete.mk (τ c)).hom (1 : k)))
/-- On contracting metrics we get back the unit. -/
contr_metric (c : C) :
(β_ (FDiscrete.obj (Discrete.mk c)) (FDiscrete.obj (Discrete.mk (τ
c)))).hom .hom

```

```

(((FDiscrete.obj (Discrete.mk c)) < (λ_ (FDiscrete.obj (Discrete.mk (τ
c))))).hom).hom
(((FDiscrete.obj (Discrete.mk c)) < ((contr.app (Discrete.mk c)) >
(FDiscrete.obj (Discrete.mk (τ c))))).hom
(((FDiscrete.obj (Discrete.mk c)) < (α_ (FDiscrete.obj (Discrete.mk (c)))
(FDiscrete.obj (Discrete.mk (τ c))) (FDiscrete.obj (Discrete.mk (τ
c))))).inv).hom
((α_ (FDiscrete.obj (Discrete.mk (c))) (FDiscrete.obj (Discrete.mk (c)))
(FDiscrete.obj (Discrete.mk (τ c))) ⊗ FDiscrete.obj (Discrete.mk (τ
c))))).hom.hom
((metric.app (Discrete.mk c)).hom (1 : k) ⊗r [k]
(metric.app (Discrete.mk (τ c)).hom (1 : k))))))
= (unit.app (Discrete.mk c)).hom (1 : k)

```

We can then let e.g.  $S : \text{TensorSpecies}$  and recover the functor  $F_D$  by  $S.F\text{Discrete}$ . The functor  $F$  is defined separately can be recovered by  $S.F$ . If  $f : X \rightarrow S.C$  is a map then a tensor is an element of  $S.F.obj (\text{OverColor.mk } f)$ . The  $\text{OverColor.mk}$  tells Lean to consider  $f$  as a function an element of the category  $T_{/C}^\times$ , or as it is written in Lean  $\text{OverColor } S.C$ .

A function  $f : \text{Fin } 2 \rightarrow S.C$ , for instance landing on  $c1, c2 \in C$  can be written in Lean as  $![c1, c2]$ , thus we can write  $S.F.obj (\text{OverColor.mk } ![c1, c2])$ .

We will see instances of this for Complex Lorentz tensors. As an example, a Lorentz tensor  $T_V^\mu$  can be written as

```

T : complexLorentzTensor.F.obj (OverColor.mk ![Color.up, Color.down])

```

## 2.2. TENSOR TREES

A tensor expression consists of a series of tensors and operations between them. For example

$$\eta_{\mu\nu} P_\sigma^\nu + V_{\mu\sigma}, \quad (9)$$

consists of a product of tensors, a contraction and an addition. Such an expression is a tensor in its own right, and we could just use the operations discussed above to define it.

However, it is useful for bridging the gap between syntax and a tensor to have a more structured way of representing such expressions. This is where tensor trees come in. Tensor trees will also be useful when it comes to proving results about tensors.

A tensor tree is essentially a tree with nodes representing tensors or operations on tensors. For example the tensor tree for the expression above is:

Since we really only care about tensors with  $X = \text{Finn}$ , tensor trees in Lean are implemented only for these.

In Lean we define a tensor tree as follows:

```

/-- A syntax tree for tensor expressions. -/
inductive TensorTree (S : TensorSpecies) : {n : ℕ} → (Fin n → S.C) → Type where
  /-- A general tensor node. -/

```

```

| tensorNode {n : ℕ} {c : Fin n → S.C} (T : S.F.obj (OverColor.mk c)) :
  TensorTree S c
/-- A node corresponding to the addition of two tensors. -/
| add {n : ℕ} {c : Fin n → S.C} : TensorTree S c → TensorTree S c → TensorTree
  S c
/-- A node corresponding to the permutation of indices of a tensor. -/
| perm {n m : ℕ} {c : Fin n → S.C} {c1 : Fin m → S.C}
  (σ : (OverColor.mk c) → (OverColor.mk c1)) (t : TensorTree S c) :
  TensorTree S c1
/-- A node corresponding to the product of two tensors. -/
| prod {n m : ℕ} {c : Fin n → S.C} {c1 : Fin m → S.C}
  (t : TensorTree S c) (t1 : TensorTree S c1) : TensorTree S (Sum.elim c c1 o
  finSumFinEquiv.symm)
/-- A node corresponding to the scalar multiple of a tensor by an element of the
  field. -/
| smul {n : ℕ} {c : Fin n → S.C} : S.k → TensorTree S c → TensorTree S c
/-- A node corresponding to negation of a tensor. -/
| neg {n : ℕ} {c : Fin n → S.C} : TensorTree S c → TensorTree S c
/-- A node corresponding to the contraction of indices of a tensor. -/
| contr {n : ℕ} {c : Fin n.succ.succ → S.C} : (i : Fin n.succ.succ) →
  (j : Fin n.succ) → (h : c (i.succAbove j) = S.τ (c i)) → TensorTree S c →
  TensorTree S (c o Fin.succAbove i o Fin.succAbove j)
/-- A node corresponding to the action of a group element on a tensor. -/
| action {n : ℕ} {c : Fin n → S.C} : S.G → TensorTree S c → TensorTree S c
/-- A node corresponding to the evaluation of an index of a tensor. -/
| eval {n : ℕ} {c : Fin n.succ → S.C} : (i : Fin n.succ) → (x : ℕ) →
  TensorTree S c →
  TensorTree S (c o Fin.succAbove i)

```

Let us give an example of what this notation means. For each  $n : \mathbb{N}$  and map  $c : \text{Fin } n \rightarrow S.C$  we get type  $\text{TensorTree } S \ c$ . This is the type of tensor trees corresponding to tensors in  $S.F.\text{obj } (\text{OverColor.mk } c)$ .

The first constructor `tensorNode` generates a tensor tree of type  $\text{TensorTree } S \ c$  from a tensor of type  $S.F.\text{obj } (\text{OverColor.mk } c)$ . The other constructors are slightly more complicated. The constructor `contr` says given an index  $i$  and  $j$ , and a tensor tree  $t$  of type  $\text{TensorTree } S \ c$ , we can get a tensor tree of type  $\text{TensorTree } S \ (c \circ \text{Fin.succAbove } i \circ \text{Fin.succAbove } j)$ .

So in Lean the expression above could be written as follows: `js: ...`

This definition of a tensor tree does not rely the actual nature of the operations involved. From a tensor tree we can define a tensor itself using a recursively defined map

```

/-- The underlying tensor a tensor tree corresponds to. -/
def tensor : ∀ {n : ℕ} {c : Fin n → S.C}, TensorTree S c → S.F.obj (OverColor.mk
c) := fun
| tensorNode t => t
| add t1 t2 => t1.tensor + t2.tensor
| perm σ t => (S.F.map σ).hom t.tensor
| neg t => - t.tensor
| smul a t => a · t.tensor
| prod t1 t2 => (S.F.map (OverColor.equivToIso finSumFinEquiv).hom).hom
((S.F.μ _ _).hom (t1.tensor ⊗ t2.tensor))
| contr i j h t => (S.contrMap _ i j h).hom t.tensor
| eval i e t => (S.evalMap i (Fin.ofNat' e Fin.size_pos')) t.tensor
| action g t => (S.F.obj (OverColor.mk _)).ρ g t.tensor

```

Note that this definition is recursive, Lean can automatically determine that this will terminate (with a little help about the number of nodes in a tensor tree). This association of a tensor tree with a tensor is not one-to-one (but it is onto). Many tensor trees can represent the same tensor, in the same way that many tensor expressions represent the same tensor.

### 2.2.1 Using Tensor trees in proofs

Tensor trees can be used in proofs for the following reason. Define a sub-tree of a tensor trees to be a node and all child nodes of that node. If  $T$  is a tensor tree and  $S$  a sub-tree of  $T$ , we can replace  $S$  in  $T$  with another tensor tree  $S'$  to get a new overall tensor-tree  $T'$ . If  $S$  and  $S'$  have the same underlying tensor, then  $T$  and  $T'$  will.

In Lean this property is encoded in a number of lemmas, for example: [js: lemma](#)

We will this at play in the example section of this paper.

## 2.3. ELABORATION

We now dicuss how we make the Lean code look similar to what we would use on pen-and-paper physics.

This is done using a two step process. Firstly, we define syntax for tensor expressions. Then we write code to turn this syntax into a tensor tree. This process is not formally defined or verified, Lean takes the outputted tensor tree as the formal object to work with.

Instead of deleving into the nitty-gritty details of how this process works under the hood, we give some examples to see how it works.

In what follows we will assume that  $T$ , and  $T_i$  etc are tensors defined as  $S.F.obj\_$  for some tensor species  $S$ . We will also assume that these tensors are defined correctly for the expressions below to make sense.

The syntax allows us to write the following

$\{T \mid \mu \ v\}^T$	tensorNode T
------------------------	--------------

for a tensor node. Here the  $\mu$  and  $v$  are free variables and it does not matter what we call them - Lean will elaborate the expression in the same way. The elaborator also knows how many indices to expect

for a tensor  $T$  and will raise an error if the wrong number are given. The  $\{_\_ \}^T$  notation is used to tell Lean that the syntax is to be treated as a tensor expression.

We can write e.g.

$\{T \mid \mu \ v\}^T$ .tensor	(tensorNode T).tensor
--------------------------------	-----------------------

to get the underlying tensor.

Note that we have not lowered or risen the indices, as one would expect from pen-and-paper notation. There is one primary reason for this, whether an index is lower or risen does not carry any information, since this information comes from the tensor itself. Also, for something like complex Lorentz tensors, there are at least three different types of upper-index.

We can extract the tensor from a tensor tree using the following syntax. If we want to evaluate an index we can put an explicit index in place of  $\mu$  or  $v$  above, for example

$\{T \mid 1 \ v\}^T$	eval 0 1 (tensorNode T)
----------------------	-------------------------

The syntax and elaboration for negation, scalar multiplication and the group action are fairly similar. For negation we have

$\{T \mid \mu \ v\}^T$	neg (tensorNode T)
------------------------	--------------------

For scalar multiplication by  $a \in k$  we have

$\{a \cdot T \mid \mu \ v\}^T$	smul a (tensorNode T)
--------------------------------	-----------------------

For the group action of  $g \in G$  on a tensor  $T$  we have

$\{g \cdot_a T \mid \mu \ v\}^T$	action g (tensorNode T)
----------------------------------	-------------------------

For the product of two tensors is also fairly simple, we have

$\{T \mid \mu \ v \otimes T_2 \mid \sigma\}^T$	prod (tensorNode T) (tensorNode T2)
--	-------------------------------------

The syntax for contraction is as one expect,

$\{T \mid \mu \ v \otimes T_2 \mid v \ \sigma\}^T$	contr 1 1 rfl (prod (tensorNode T) (tensorNode T2))
--	---

On the RHS here the first argument (1) of contr is the index of the first  $v$  on the LHS, the second argument (also 1) is the second index. The rfl is a proof that the colors of the two contracted indices are actually dual to one another. If they are not, this proof will fail and the elaborator will complain. It will also complain if more then two indices are trying to be contracted. Although, this depends on where exactly the indices sit in the expression. For example

$\{T \mid \mu \ v \otimes T_2 \mid v \ v\}^T$	(prod (tensorNode T) (contr 0 0 rfl (tensorNode T2)))
---	---

works fine because the inner contraction is done before the product.

We now turn to addition. Our syntax allows for e.g.  $\{T \mid \mu \ v + T_2 \mid \mu \ v\}^T$  and also  $\{T \mid \mu \ v + T_2 \mid v \ \mu\}^T$ , provided of course that the indices are of the correct color (which Lean will check). The elaborator handles both these cases, and generalizations thereof by adding a permutation node. Thus we have

$\{T \mid \mu \ v + T_2 \mid \mu \ v\}^T$	add (tensorNode T) (perm _ (tensorNode T2))
---	---

where here the  $\_$  is a placeholder for the permutation, and in this case will be trivial, but for

$\{T \mid \mu \ v + T2 \mid v \ \mu\}^T$	<code>add (tensorNode T) (perm _ (tensorNode T2))</code>
--	--

it will be the permutation for the two identities.

Despite not forming part of a node in our tensor tree, we also give syntax for equality. This is done in a very similar way to addition, with the addition of a permutation node to account for e.g. the fact that  $T_{\mu\nu} = T_{\nu\mu}$ .

$\{T \mid \mu \ v = T2 \mid v \ \mu\}^T$	<code>(tensorNode T).tensor = (perm _ (tensorNode T2)).tensor</code>
--	--

Note the use of the `.tensor` to extract the tensor from the tensor tree, it does not really mean much to ask for equality of the tensor trees themselves.

### 3. EXAMPLES

We give two examples in this section. The first example is a simple theorem involving index notation and tensor trees. We will show here, in rather explicit detail, how we can manipulate tensor trees to solve such theorems. The second example we shall give will show a number of definitions in HepLean concerning index notation. Here we will not give so much detail, the point rather being to show the reader the broad use of our construction.

#### 3.1. EXAMPLE 1: SYMMETRIC AND ANTI-SYMMETRIC TENSOR

If  $A^{\mu\nu}$  is an anti-symmetric tensor and  $S_{\mu\nu}$  and  $S$  is a symmetric tensor, then it is true that  $A^{\mu\nu}S_{\mu\nu} = -A^{\mu\nu}S_{\nu\mu}$ . In Lean this result, and it's lemma are written as follows:

```
lemma antiSymm_contr_symm
  {A : complexLorentzTensor.F.obj (OverColor.mk ![Color.up, Color.up])}
  {S : complexLorentzTensor.F.obj (OverColor.mk ![Color.down, Color.down])}
  (hA : {A |  $\mu \ v = - (A | v \ \mu)\}^T$ ) (hs : {S |  $\mu \ v = S | v \ \mu\}^T$ ) :
  {A |  $\mu \ v \otimes S | \mu \ v = - A | \mu \ v \otimes S | \mu \ v\}^T := by
conv =>
  lhs
  rw [contr_tensor_eq <| contr_tensor_eq <| prod_tensor_eq_fst <| hA]
  rw [contr_tensor_eq <| contr_tensor_eq <| prod_tensor_eq_snd <| hs]
  rw [contr_tensor_eq <| contr_tensor_eq <| prod_perm_left _ _ _]
  rw [contr_tensor_eq <| contr_tensor_eq <| perm_tensor_eq <| prod_perm_right _ _
    _]
  rw [contr_tensor_eq <| contr_tensor_eq <| perm_perm _ _]
  rw [contr_tensor_eq <| perm_contr_congr 1 2]
  rw [perm_contr_congr 0 0]
  rw [perm_tensor_eq <| contr_contr _ _]
  rw [perm_perm]
  rw [perm_tensor_eq <| contr_tensor_eq <| contr_tensor_eq <| neg_fst_prod _ _]
  rw [perm_tensor_eq <| contr_tensor_eq <| neg_contr _]
  rw [perm_tensor_eq <| neg_contr _]
  apply perm_congr _ rfl
  decide$ 
```

Let us break this down. The statements

```
{A : complexLorentzTensor.F.obj (OverColor.mk ![Color.up, Color.up])}
{S : complexLorentzTensor.F.obj (OverColor.mk ![Color.down, Color.down])}
```

are simply defining  $A$  and  $S$  to be tensors of type  $A^{\mu\nu}$  and  $S_{\mu\nu}$  respectively. This agrees with the notation set out in §?? [js: ref](#).


The parameter `hA` is stating that  $A$  is anti-symmetric. Expanded in terms of tree diagrams we have

$hA : \{A \mid \mu \nu = - (A \mid \nu \mu)\}^T$	
Description: The tensor $A$ is anti-symmetric.	

Similarly, the parameter `hS` is stating that  $S$  is symmetric. Expanded in terms of tree diagrams

$hS : \{S \mid \mu \nu = S \mid \nu \mu\}^T$	
Description: The tensor $S$ is symmetric.	

The line  $\{A \mid \mu \nu \otimes S \mid \mu \nu = - A \mid \mu \nu \otimes S \mid \mu \nu\}^T$  is the statment we are trying to prove. In terms of tree diagrams it says that




The `perm` here actually does nothing, but is included by Lean.

The lines of the proof in the `conv` block are manipulations of the tensor tree on the LHS of the equation. The `rw` tactic is used to rewrite the tensor tree using the various lemmas. We go through each step in turn.




```
rw [contr_tensor_eq <| contr_tensor_eq
<| prod_tensor_eq_fst <| hA]
```

Description: Replace the node A with the RHS of hA.




```
rw [contr_tensor_eq <| contr_tensor_eq
<| prod_tensor_eq_snd <| hS]
```

Description: Replace the node S with the RHS of hS.




```
rw [contr_tensor_eq <| contr_tensor_eq
<| prod_perm_left _ _ _]
```

Description: Move the left permutation through the product.




```
rw [contr_tensor_eq <| contr_tensor_eq
<| perm_tensor_eq <| prod_perm_right _ _
_ _]
```

Description: Move the right permutation through the product.




```
rw [contr_tensor_eq <| contr_tensor_eq
<| perm_perm _ _ _]
```

Description: Combine the two permutations (using functoriality).




```
rw [contr_tensor_eq <| perm_contr_congr
1 2]
```

Description: Move the permutation through the contraction. And simplify the contraction indices to 1 and 2 (Lean will check if this is correct).




```
rw [perm_contr_congr 0 0]
```

Description: Move the permutation through the contraction. And simplify the contraction indices to 0 and 0 (Lean will check if this is correct).




```
rw [perm_tensor_eq <| contr_contr _ _ _]
```

Description: Swap the two contractions. This introduces a permutation.




```
rw [perm_perm]
```

Description: Combine the two permutations.




```
rw [perm_tensor_eq <| contr_tensor_eq <|
  contr_tensor_eq <| negfst_prod _ _]
```

Description: Move the negation through the product.




```
rw [perm_tensor_eq <| contr_tensor_eq <|
  neg_contr _]
```

Description: Move the negation through the first contraction.



```
rw [perm_tensor_eq <| neg_contr _]
```

Description: Move the negation through the second contraction.



The remainder of the proof `apply perm_congr _ rfl` and decide that the tensor trees on the LHS and RHS are actually equal.

### 3.2. EXAMPLE 2: PAULI MATRICES AND BISPINORS

Using the formalism we have set up thus far it is possible to define Pauli matrices and bispinors as complex Lorentz tensors.

The pauli matrices appear in HepLean as follows

```

/-- The Pauli matrices as the complex Lorentz tensor ' $\sigma^{\mu\alpha\dot{\beta}}$ '. -/
def pauliContr := {PauliMatrix.asConsTensor |  $\nu \alpha \beta$ }T.tensor

/-- The Pauli matrices as the complex Lorentz tensor ' $\sigma_{\mu\alpha\dot{\beta}}$ '. -/
def pauliCo := { $\eta'$  |  $\mu \nu \otimes \text{pauliContr}$  |  $\nu \alpha \beta$ }T.tensor

/-- The Pauli matrices as the complex Lorentz tensor ' $\sigma_{\mu\alpha\dot{\beta}}$ '. -/
def pauliCoDown := {pauliCo |  $\mu \alpha \beta \otimes \epsilon_L$ ' |  $\alpha \alpha' \otimes \epsilon_R$ ' |  $\beta \beta'$ }T.tensor

/-- The Pauli matrices as the complex Lorentz tensor ' $\sigma^{\mu\alpha\dot{\beta}}$ '. -/
def pauliContrDown := {pauliContr |  $\mu \alpha \beta \otimes \epsilon_L$ ' |  $\alpha \alpha' \otimes \epsilon_R$ ' |  $\beta \beta'$ }T.tensor

```

The first of these definitions depends on `PauliMatrix.asConsTensor` which is defined as using an explicit basis expansion.

In these expressions we have the appearance of metrics  $\eta'$ , is the metric usually written as  $\eta_{\mu\nu}$ , `js:` etc.

With these we can define bispinors

```

/-- A bispinor ' $p^{aa}$ ' created from a lorentz vector ' $p^{\mu}$ '. -/
def contrBispinorUp (p : complexContr) :=
  {pauliCo |  $\mu \alpha \beta \otimes p$  |  $\mu$ }T.tensor

/-- A bispinor ' $p_{aa}$ ' created from a lorentz vector ' $p^{\mu}$ '. -/
def contrBispinorDown (p : complexContr) :=
  { $\epsilon_L$ ' |  $\alpha \alpha' \otimes \epsilon_R$ ' |  $\beta \beta' \otimes \text{contrBispinorUp } p$  |  $\alpha \beta$ }T.tensor

/-- A bispinor ' $p^{aa}$ ' created from a lorentz vector ' $p_{\mu}$ '. -/
def coBispinorUp (p : complexCo) := {pauliContr |  $\mu \alpha \beta \otimes p$  |  $\mu$ }T.tensor

/-- A bispinor ' $p_{aa}$ ' created from a lorentz vector ' $p_{\mu}$ '. -/
def coBispinorDown (p : complexCo) :=
  { $\epsilon_L$ ' |  $\alpha \alpha' \otimes \epsilon_R$ ' |  $\beta \beta' \otimes \text{coBispinorUp } p$  |  $\alpha \beta$ }T.tensor

```

Here `complexContr` and `complexCo` are complex contravariant and covariant Lorentz vectors. Lean knows to treat these as tensors when they appear in tensor expressions.

Using these definitions we can start to prove results about the pauli matrices and bispinors. These proofs rely on essentially the sorts of manipulations in the last section, although in some cases we expand tensors in terms of a basis and use rules about how the basis interacts with the operations in a tensor tree.

Examples of things we have proven range

```

lemma coBispinorDown_eq_pauliContrDown_contr (p : complexCo) :
  {coBispinorDown p |  $\alpha \beta$  = pauliContrDown |  $\mu \alpha \beta \otimes p$  |  $\mu$ }T := by

```

the proof of which is an application of associativity of the tensor product, and appropriately shuffling around of the contractions.

To more complicated results such as

```

/-- The statement that ' $\eta_{\{\mu\nu\}} \sigma^{\{\mu \alpha \text{ dot } \beta\}} \sigma^{\{v \alpha' \text{ dot } \beta'\}} = 2 \varepsilon^{\{\alpha\alpha'\}} \varepsilon^{\{\text{dot } \beta \text{ dot } \beta'\}}$ '. -/
theorem pauliCo_contr_pauliContr :
  {pauliCo |  $v \alpha \beta \otimes \text{pauliContr} \mid v \alpha' \beta' = 2 \cdot \varepsilon_L \mid \alpha \alpha' \otimes \varepsilon_R \mid \beta \beta'\}^T := \text{by}$ 
```

#### 4. FUTURE WORK

The scale of formalizing all results regarding index notation is a task that surpasses the capacity of any single individual. Inspired by the Lean community’s blueprint projects, we have added to HepLean informal lemmas related to index notation and tensors. An example of such is

```

informal_lemma coBispinorUp_eq_metric_contr_coBispinorDown where
  math :≈ "{coBispinorUp p |  $\alpha \beta = \varepsilon_L \mid \alpha \alpha' \otimes \varepsilon_R \mid \beta \beta' \otimes \text{coBispinorDown p} \mid \alpha$   

    ,  $\beta' \}$ "
  proof :≈ "Expand 'coBispinorDown' and use fact that metrics contract to the  

    identity."
  deps :≈ ["coBispinorUp", "coBispinorDown", "leftMetric", "rightMetric"]
```

The lemmas resource that we hope others—or even automated systems—will formalize in the future.

As demonstrated in our earlier examples, manipulating tensor expressions can involve tedious calculations, especially when dealing with directly with tensor trees. In the future, we intend to automate many of these routine steps by developing suitable tactics within Lean. We are optimistic that the structured nature of tensor trees will lend itself well to such automation, thereby streamlining computations and enhancing the efficiency of formal proofs involving index notation and tensor species.

There are two primary directions in which we can extend the concepts presented in this work. First, we could incorporate the spinor-helicity formalism, which is used in the study of scattering amplitudes. Second, we could extend our approach to encompass tensor *fields*, their derivatives etc. We do not anticipate any insurmountable challenges in pursuing these extensions. They represent promising avenues for future research and have the potential to significantly enhance the utility of formal methods in physics.

#### REFERENCES